
  0.4.2 (2024-07-11)



  
    
  
  Documentation


	this release is a documentation update and does not contain any new features or bug fixes


  
    
  
  0.4.1 (2024-07-10)



  
    
  
  Documentation


	this release is a documentation update and does not contain any new features or bug fixes


  
    
  
  0.4.0 (2024-07-07)



  
    
  
  ⚠ BREAKING CHANGES


	store parameters as keywords instead of atoms to retain more information (#10)


  
    
  
  Code Refactoring


	store parameters as keywords instead of atoms to retain more information (#10) (df5b17e)


  
    
  
  Features


	add doc_fields guide (#11)


  
    
  
  0.3.0 (2024-07-03)



  
    
  
  ⚠ BREAKING CHANGES


	remove accessible plugin, add it to guides


  
    
  
  Bug Fixes


	macro capture will capture import context (9c6f340)


  
    
  
  Code Refactoring


	remove accessible plugin, add it to guides (204cb0a)


  
    
  
  0.2.0 (2024-06-30)



  
    
  
  Features


	register plugins globally by configuration (4767e62)


  
    
  
  0.1.5 (2024-06-29)


This release is a documentation update and
does not contain any new features or bug fixes.

  
    
  
  0.1.4 (2024-06-29)


This release is a documentation update and
does not contain any new features or bug fixes.

  
    
  
  0.1.3 (2024-06-29)



  
    
  
  Features


	doc: add Migrate from typed_struct guide


  
    
  
  Bug Fixes


	fix: exclude key from @enforce_keys, if :default is set and :enforce is true


  
    
  
  0.1.2 (2024-06-29)



  
    
  
  Bug Fixes


	avoid enforcing the field when default is unset and enforce is false


  
    
  
  0.1.1 (2024-06-28)



  
    
  
  Bug Fixes


	if default is set, then this field is enforced (#2)


  
    
  
  0.1.0 (2024-06-28)



  
    
  
  Features


First release

    Migrate from typed_struct - TypedStructor v0.4.2
    
    

    


  
  

    
Migrate from <code class="inline">typed_struct</code>
    

This package(typed_structor) keeps the same API as typed_struct.
It is a drop-in replacement for typed_struct.

  
    
  
  Migration


	Replace typed_struct with typed_structor in your mix.exs file.

   defp deps do
     [
       # ...deps
-      {:typed_struct, "~> 0.3.0"},
+      {:typed_structor, "~> 0.4"},
     ]
   end
	Run mix do deps.unlock --unused, deps.get, deps.clean --unused to fetch the new dependency.
	Replace TypedStruct with TypedStructor in your code.

 defmodule User do
-  use TypedStruct
+  use TypedStructor
 
-  typedstruct do
+  typed_structor do
     field :id, pos_integer()
     field :name, String.t()
     field :age, non_neg_integer()
   end
 end
	That's it!



  

  
    
    Introduction - TypedStructor v0.4.2
    
    

    


  
  

    
Introduction to the plugin system
    

For more customization, TypedStructor provides a plugin system
that allows you to extend the functionality of the library.
This is useful when you want to extract some common logic into a separate module.
See TypedStructor.Plugin for how to create a plugin.
This library comes with a few built-in plugins, and we don't like to
implement more built-in plugins, but instead, we encourage you to create your own plugins.
We provide some example plugins that you can use as a reference, or copy-paste.
Plugin examples:
	Registering plugins globally
	Implement Access behavior
	Implement reflection functions
	Type Only on Ecto Schema
	Add primary key and timestamps types to your Ecto schema
	Derives the Jason.Encoder for struct
	Derives the Enumerable for struct



  

  
    
    Registering plugins globally - TypedStructor v0.4.2
    
    

    


  
  

    
Registering plugins globally
    

TypedStructor allows you to register plugins globally, so you don't have to specify them in each struct.
This is useful when you want to apply the same plugin to all modules that use TypedStructor.
Global plugins are applied to all modules
The global registered plugins is applied to all modules that use TypedStructor.
That means any dependency or library that uses TypedStructor will also be affected.
If you want to apply the plugin to specific modules, you can determine the
module name or other conditions in the plugin implementation.


  
    
  
  Usage


To register a plugin globally, you can add it to the :plugins key in the :typed_structor app configuration.
config :typed_structor, plugins: [MyPlugin, {MyPluginWithOpts, [foo: :bar]}]

  
    
  
  How to opt-out the plugin conditionally


The most common way to opt-out a plugin is to determine the module name.
defmodule MyPlugin do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    quote do
      if unquote(__MODULE__).__opt_in__?(__MODULE__) do
        # do something
      end

      unquote(definition)
    end
  end

  @impl TypedStructor.Plugin
  defmacro after_definition(_definition, _opts) do
    quote do
      if unquote(__MODULE__).__opt_in__?(__MODULE__) do
        # do something
      end
    end
  end

  def __opt_in__?(module) when is_atom(module) do
    # Opt-in only for schemas under MyApp.Schemas
    String.starts_with?(Atom.to_string(module), "MyApp.Schemas")
  end
end

Implement <code class="inline">Access</code> behavior
    

Sometimes, you may want to use the get_in/2 and update_in/3 functions
with your structs. This guide demonstrates how to effortlessly implement
the Access behavior for your structs using a plugin.

  
    
  
  Implementation


Destructive operations
These operations are not allowed for the struct:
	update :__struct__ key
	pop a key

The functions will raise an ArgumentError if called.
To enable these functionalities, override the get_and_update/3 and pop/2 functions.

defmodule Guides.Plugins.Accessible do
  @moduledoc """
  This plugin implements the `Access` behavior for the struct
  by delegating the `fetch/2`, `get_and_update/3`, and `pop/2`
  functions to the `Map` module.

  > #### Destructive operations  {: .warning}
  > These operations are not allowed for the struct:
  > * update `:__struct__` key
  > * pop a key
  >
  > The functions will raise an `ArgumentError` if called.
  > To enable these functionalities, override the `get_and_update/3` and `pop/2` functions.

  ## Usage

      typed_structor do
        plugin TypedStructor.Plugins.Accessible

        # fields
      end
  """

  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro after_definition(_definition, _opts) do
    quote do
      @behaviour Access

      @impl Access
      defdelegate fetch(term, key), to: Map

      @impl Access
      def get_and_update(data, :__struct__, _function) do
        raise ArgumentError,
              "Cannot update `:__struct__` key." <>
                "To enable this functionality, implement `Access.get_and_update/3` for #{inspect(__MODULE__)} to override this behaviour."
      end

      defdelegate get_and_update(data, key, function), to: Map

      @impl Access
      def pop(data, key) do
        raise ArgumentError,
              "Cannot pop `#{inspect(key)}` key.\n" <>
                "To enable this functionality, implement `Access.pop/2` for #{inspect(__MODULE__)} to override this behaviour."
      end

      @defoverridable Access
    end
  end
end

  
    
  
  Usage


defmodule User do
  use TypedStructor

  typed_structor do
    plugin Guides.Plugins.Accessible

    field :name, String.t()
    field :age, integer()
  end
end
iex> user = %User{name: "Phil", age: 20}
%User{name: "Phil", age: 20}
iex> get_in(user, [:name])
"Phil"
iex> put_in(user, [:name], "phil")
%User{name: "phil", age: 20}

Implement reflection functions
    

Define a plugin that generates reflection functions that
can be used to access the fields and parameters of a struct.

  
    
  
  Implement


defmodule Guides.Plugins.Reflection do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro after_definition(definition, _opts) do
    quote bind_quoted: [definition: definition] do
      fields = Enum.map(definition.fields, &Keyword.fetch!(&1, :name))

      enforced_fields =
        definition.fields
        |> Stream.filter(&Keyword.get(&1, :enforce, false))
        |> Stream.map(&Keyword.fetch!(&1, :name))
        |> Enum.to_list()

      def __typed_structor__(:fields), do: unquote(fields)
      def __typed_structor__(:parameters), do: Enum.map(unquote(definition.parameters), &Keyword.fetch!(&1, :name))
      def __typed_structor__(:enforced_fields), do: unquote(enforced_fields)

      for field <- definition.fields do
        name = Keyword.fetch!(field, :name)
        type = field |> Keyword.fetch!(:type) |> Macro.escape()

        def __typed_structor__(:type, unquote(name)), do: unquote(type)
        def __typed_structor__(:field, unquote(name)), do: unquote(Macro.escape(field))
      end
    end
  end
end

  
    
  
  Usage


defmodule User do
  use TypedStructor

  typed_structor do
    plugin Guides.Plugins.Reflection

    parameter :age

    field :name, String.t(), enforce: true
    field :age, age, default: 20
  end
end

defmodule MyApp do
  use TypedStructor

  typed_structor module: User, enforce: true do
    plugin Guides.Plugins.Reflection

    field :name, String.t()
    field :age, integer()
  end
end
iex> User.__typed_structor__(:fields)
[:name, :age]
iex> User.__typed_structor__(:parameters)
[:age]

iex> MyApp.User.__typed_structor__(:enforced_fields)
[:name, :age]
iex> Macro.to_string(User.__typed_structor__(:type, :age))
"age"

Add fields docs to the <code class="inline">@typedoc</code>
    

For enhanced readability, it’s often beneficial to document the fields
and parameters of a struct. Incorporating documentation can significantly
aid both your co-workers and your future self by providing clear,
accessible information.
Below is the final code generated by the plugin, which includes comprehensive
documentation for the User struct:
defmodule User do
  @typedoc """
  @type t(age) :: %User{age: age | nil, name: String.t() | nil}

  This is a user struct.

  ## Parameters

  Name | Description
  :age | The age parameter.

  ## Fields

  Name  | Type             | Description
  :name | String.t() | nil | The name of the user.
  :age  | age | nil        | The age of the user.
  """
  @type t(age) :: %__MODULE__{name: String.t() | nil, age: age | nil}

  defstruct [:name, :age]
end

  
    
  
  Implement


defmodule Guides.Plugins.DocFields do
  @moduledoc """
  The `DocFields` plugin generates documentation for fields and parameters.
  Simply add the `:doc` option to the `field` and `parameter` macros to document them.

  ## Example

      use TypedStructor

      typed_structor do
        @typedoc \"""
        This is a user struct.
        \"""
        plugin Guides.Plugins.DocFields

        parameter :age, doc: "The age parameter."

        field :name, String.t(), doc: "The name of the user."
        field :age, age, doc: "The age of the user."
      end

  This will generate the following documentation for you:

      @typedoc \"""
      This is a user struct.


      ## Parameters

      | Name | Description |
      |------|-------------|
      |`:age` | The age parameter.|


      ## Fields

      | Name | Type | Description |
      |------|------|-------------|
      |`:name` | `String.t() \| nil` | The name of the user.|
      |`:age` | `age \| nil` | The age of the user.|
      \"""

      @type t(age) :: %User{age: age | nil, name: String.t() | nil}
  """

  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    quote do
      @typedoc unquote(__MODULE__).__generate_doc__(unquote(definition), @typedoc)

      unquote(definition)
    end
  end

  def __generate_doc__(_definition, false), do: nil

  def __generate_doc__(definition, typedoc) do
    parameters =
      Enum.map(definition.parameters, fn parameter ->
        name = Keyword.fetch!(parameter, :name)
        doc = Keyword.get(parameter, :doc, "*not documented*")

        ["`#{inspect(name)}`", doc]
      end)

    parameters_docs =
      if length(parameters) > 0 do
        """
        ## Parameters

        | Name | Description |
        |------|-------------|
        #{join_rows(parameters)}
        """
      end

    fields =
      Enum.map(definition.fields, fn field ->
        name = Keyword.fetch!(field, :name)

        type = Keyword.fetch!(field, :type)

        type =
          if Keyword.get(field, :enforce, false) or Keyword.has_key?(field, :default) do
            Macro.to_string(type)
          else
            # escape `|`
            "#{Macro.to_string(type)} \\| nil"
          end

        doc = Keyword.get(field, :doc, "*not documented*")

        ["`#{inspect(name)}`", "`#{type}`", doc]
      end)

    fields_docs =
      if length(fields) > 0 do
        """
        ## Fields

        | Name | Type | Description |
        |------|------|-------------|
        #{join_rows(fields)}
        """
      end

    [parameters_docs, fields_docs]
    |> Enum.reject(&is_nil/1)
    |> case do
      [] ->
        typedoc

      docs ->
        """
        #{typedoc}

        #{Enum.join(docs, "\n\n")}
        """
    end
  end

  defp join_rows(rows) do
    Enum.map_join(rows, "\n", fn row -> "|" <> Enum.join(row, " | ") <> "|" end)
  end
end

  
    
  
  Usage


defmodule User do
  @moduledoc false

  use TypedStructor

  typed_structor do
    @typedoc """
    This is a user struct.
    """
    plugin Guides.Plugins.DocFields

    parameter :age, doc: "The age parameter."

    field :name, String.t(), doc: "The name of the user."
    field :age, age, doc: "The age of the user."
  end
end
iex> t User.t
@type t(age) :: %User{age: age | nil, name: String.t() | nil}

This is a user struct.

## Parameters

Name | Description
:age | The age parameter.

## Fields

Name  | Type             | Description
:name | String.t() | nil | The name of the user.
:age  | age | nil        | The age of the user.

Type Only on Ecto Schema
    

Ecto is a great library for working with both databases and data validation.
However, it has its own way of defining schemas and fields,
which results a struct but without type definitions.
This plugin automatically disables struct creation for Ecto schemas.

  
    
  
  Implementation


It uses the TypedStructor.Plugin.before_definition/2 callback to determine if the module is an Ecto schema
by checking the @ecto_fields module attribute. If it is, the :define_struct option is
set to false to prevent struct creation.
Here is the plugin(feel free to copy and paste):
defmodule Guides.Plugins.TypeOnlyOnEctoSchema do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    quote do
      if Module.has_attribute?(__MODULE__, :ecto_fields) do
        Map.update!(unquote(definition), :options, fn opts ->
          Keyword.put(opts, :define_struct, false)
        end)
      else
        unquote(definition)
      end
    end
  end
end

  
    
  
  Usage


To use this plugin, you can add it to the typed_structor block like this:
defmodule MyApp.User do
  use TypedStructor
  use Ecto.Schema

  typed_structor do
    plugin Guides.Plugins.TypeOnlyOnEctoSchema

    field :id, integer(), enforce: true
    field :name, String.t()
    field :age, integer(), enforce: true # There is always a non-nil value
  end

  schema "source" do
    field :name, :string
    field :age, :integer, default: 20
  end
end

  
    
  
  Registering the plugin globally


config :typed_structor, plugins: [Guides.Plugins.TypeOnlyOnEctoSchema]
Note that the plugin is applied to all modules that use TypedStructor,
you can opt-out by determining the module name or other conditions.
Let's change the plugin to only apply to modules from the MyApp namespace(feel free to copy and paste):
defmodule Guides.Plugins.TypeOnlyOnEctoSchema do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    quote do
      # Check if the module is from the MyApp namespace
      with "MyApp" <- __MODULE__ |> Module.split() |> hd(),
           true <- Module.has_attribute?(__MODULE__, :ecto_fields) do
        Map.update!(unquote(definition), :options, fn opts ->
          Keyword.put(opts, :define_struct, false)
        end)
      else
        _otherwise -> unquote(definition)
      end
    end
  end
end
Now you can use typed_structor without registering the plugin explicitly:
 defmodule MyApp.User do
   use TypedStructor
   use Ecto.Schema

   typed_structor do
-    plugin Guides.Plugins.TypeOnlyOnEctoSchema
 
     field :id, integer(), enforce: true
     field :name, String.t()
     field :age, integer(), enforce: true
   end

   schema "source" do
     field :name, :string
     field :age, :integer, default: 20
   end
 end


  

  
    
    Add primary key and timestamps types to your Ecto schema - TypedStructor v0.4.2
    
    

    


  
  

    
Add primary key and timestamps types to your Ecto schema
    


  
    
  
  Implementation


This plugin use TypedStructor.Plugin.before_definition/2 callback to
inject the primary key and timestamps fields to the type definition.
defmodule Guides.Plugins.PrimaryKeyAndTimestamps do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    quote do
      Map.update!(unquote(definition), :fields, fn fields ->
        # Assume that the primary key is an integer
        primary_key = [name: :id, type: quote(do: integer()), enforce: true]

        # Add two default timestamps
        timestamps = [
          [name: :inserted_at, type: quote(do: NaiveDateTime.t()), enforce: true],
          [name: :updated_at, type: quote(do: NaiveDateTime.t()), enforce: true]
        ]

        [primary_key | fields] ++ timestamps
      end)
    end
  end
end

  
    
  
  Usage


defmodule User do
  use TypedStructor
  use Ecto.Schema

  # disable struct creation or it will conflict with the Ecto schema
  typed_structor define_struct: false do
    # register the plugin
    plugin Guides.Plugins.PrimaryKeyAndTimestamps

    field :name, String.t()
    field :age, integer(), enforce: true # There is always a non-nil value
  end

  schema "source" do
    field :name, :string
    field :age, :integer, default: 20

    timestamps()
  end
end
If you want to apply this plugin conditionally, refer to the Registering plugins globally section.

Derives the <code class="inline">Jason.Encoder</code> for <code class="inline">struct</code>
    

We use the TypedStructor.Plugin.after_definition/2 callback to
generate the Jason.Encoder implementation for the struct.

  
    
  
  Implementation


defmodule Guides.Plugins.Jason do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro after_definition(definition, _opts) do
    quote bind_quoted: [definition: definition] do
      # Extract the field names
      keys = Enum.map(definition.fields, &Keyword.fetch!(&1, :name))

      defimpl Jason.Encoder do
        def encode(value, opts) do
          Jason.Encode.map(Map.take(value, unquote(keys)), opts)
        end
      end
    end
  end
end

  
    
  
  Usage


defmodule User do
  use TypedStructor

  typed_structor enforce: true do
    plugin Guides.Plugins.Jason

    field :name, String.t()
  end
end
After compiled, you got:
iex> Jason.encode(%User{name: "Phil"})
{:ok, "{\"name\":\"Phil\"}"}

Derives the <code class="inline">Enumerable</code> for <code class="inline">struct</code>
    

We use the TypedStructor.Plugin.after_definition/2 callback to
generate the Enumerable implementation for the struct.
We implement Enumerable callbacks exclusively using the fields that are defined.
Let's start!

  
    
  
  Implementation


defmodule Guides.Plugins.Enumerable do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro after_definition(definition, _opts) do
    quote bind_quoted: [definition: definition] do
      keys = Enum.map(definition.fields, &Keyword.fetch!(&1, :name))

      defimpl Enumerable do
        def count(enumerable), do: {:ok, Enum.count(unquote(keys))}
        def member?(enumerable, element), do: {:ok, Enum.member?(unquote(keys), element)}

        def reduce(enumerable, acc, fun) do
          # The order of fields is guaranteed to align with the sequence in which they are defined.
          unquote(keys)
          |> Enum.map(fn key -> {key, Map.fetch!(enumerable, key)} end)
          |> Enumerable.List.reduce(acc, fun)
        end

        # We don't support this
        def slice(_enumerable), do: {:error, __MODULE__}
      end
    end
  end
end

  
    
  
  Usage


defmodule User do
  use TypedStructor

  typed_structor enforce: true do
    plugin Guides.Plugins.Enumerable

    field :name, String.t()
    field :age, Integer.t()
  end
end
iex> user = %User{name: "Phil", age: 20}
%User{name: "Phil", age: 20}
# the order of fields is deterministic
iex> Enum.map(user, fn {key, _value} -> key end)
[:name, :age]
# we got a deterministic ordered Keyword list
iex> Enum.to_list(user)
[name: "Phil", age: 20]
Bonus
Additionally, we gain the bonus of having a deterministic order of fields,
determined by the sequence in which they are defined.

TypedStructor 
    



      
TypedStructor is a library for defining structs with types effortlessly.
(This library is a rewritten version of TypedStruct because it is no longer actively maintained.)

  
    
  
  Installation


Add :typed_structor to the list of dependencies in mix.exs:
def deps do
  [
    {:typed_structor, "~> 0.4"}
  ]
end
Add :typed_structor to your .formatter.exs file
[
  # import the formatter rules from `:typed_structor`
  import_deps: [..., :typed_structor],
  inputs: [...]
]

  
    
  
  Usage



  
    
  
  General usage


To define a struct with types, use TypedStructor,
and then define fields under the TypedStructor.typed_structor/2 macro,
using the TypedStructor.field/3 macro to define each field.
defmodule User do
  # use TypedStructor to import the `typed_structor` macro
  use TypedStructor

  typed_structor do
    # Define each field with the `field` macro.
    field :id, pos_integer()

    # set a default value
    field :name, String.t(), default: "Unknown"

    # enforce a field
    field :age, non_neg_integer(), enforce: true
  end
end
This is equivalent to:
defmodule User do
  defstruct [:id, :name, :age]

  @type t() :: %__MODULE__{
    id: pos_integer() | nil,
    # Note: The 'name' can not be nil, for it has a default value.
    name: String.t(),
    age: non_neg_integer()
  }
end
Check TypedStructor.typed_structor/2 and TypedStructor.field/3 for more information.

  
    
  
  Options


You can also generate an opaque type for the struct,
even changing the type name:
defmodule User do
  use TypedStructor

  typed_structor type_kind: :opaque, type_name: :profile do
    field :id, pos_integer()
    field :name, String.t()
    field :age, non_neg_integer()
  end
end
This is equivalent to:
defmodule User do
  use TypedStructor

  defstruct [:id, :name, :age]

  @opaque profile() :: %__MODULE__{
    id: pos_integer() | nil,
    name: String.t() | nil,
    age: non_neg_integer() | nil
  }
end
Type parameters also can be defined:
defmodule User do
  use TypedStructor

  typed_structor do
    parameter :id
    parameter :name

    field :id, id
    field :name, name
    field :age, non_neg_integer()
  end
end
becomes:
defmodule User do
  @type t(id, name) :: %__MODULE__{
    id: id | nil,
    name: name | nil,
    age: non_neg_integer() | nil
  }

  defstruct [:id, :name, :age]
end
If you prefer to define a struct in a submodule, you can use
the module option with TypedStructor. This allows you to
encapsulate the struct definition within a specific submodule context.
Consider this example:
defmodule User do
  use TypedStructor

  # `%User.Profile{}` is generated
  typed_structor module: Profile do
    field :id, pos_integer()
    field :name, String.t()
    field :age, non_neg_integer()
  end
end
When defining a struct in a submodule, the typed_structor block
functions similarly to a defmodule block. Therefore,
the previous example can be alternatively written as:
defmodule User do
  defmodule Profile do
    use TypedStructor

    typed_structor do
      field :id, pos_integer()
      field :name, String.t()
      field :age, non_neg_integer()
    end
  end
end
Furthermore, the typed_structor block allows you to
define functions, derive protocols, and more, just
as you would within a defmodule block. Here's a example:
defmodule User do
  use TypedStructor

  typed_structor module: Profile, define_struct: false do
    @derive {Jason.Encoder, only: [:email]}
    field :email, String.t()

    use Ecto.Schema
    @primary_key false

    schema "users" do
      Ecto.Schema.field(:email, :string)
    end

    import Ecto.Changeset

    def changeset(%__MODULE__{} = user, attrs) do
      user
      |> cast(attrs, [:email])
      |> validate_required([:email])
    end
  end
end
Now, you can interact with these structures:
iex> User.Profile.__struct__()
%User.Profile{__meta__: #Ecto.Schema.Metadata<:built, "users">, email: nil}
iex> Jason.encode!(%User.Profile{})
"{\"email\":null}"
iex> User.Profile.changeset(%User.Profile{}, %{"email" => "my@email.com"})
#Ecto.Changeset<
  action: nil,
  changes: %{email: "my@email.com"},
  errors: [],
  data: #User.Profile<>,
  valid?: true
>

  
    
  
  Documentation


To add a @typedoc to the struct type, just add the attribute in the typed_structor block:
typed_structor do
  @typedoc "A typed user"

  field :id, pos_integer()
  field :name, String.t()
  field :age, non_neg_integer()
end
You can also document submodules this way:
typedstructor module: Profile do
  @moduledoc "A user profile struct"
  @typedoc "A typed user profile"

  field :id, pos_integer()
  field :name, String.t()
  field :age, non_neg_integer()
end

  
    
  
  Plugins


TypedStructor offers a plugin system to enhance functionality.
For details on creating a plugin, refer to the TypedStructor.Plugin module.
Here is a example of Guides.Plugins.Accessible plugin to define Access behavior for the struct.
defmodule User do
  use TypedStructor

  typed_structor do
    plugin Guides.Plugins.Accessible

    field :id, pos_integer()
    field :name, String.t()
    field :age, non_neg_integer()
  end
end

user = %User{id: 1, name: "Phil", age: 20}
get_in(user, [:name]) # => "Phil"
Plugins guides
Here are some Plugin Guides
for creating your own plugins. Please check them out
and feel free to copy-paste the code.

TypedStructor.Definition 
    



      
The definition struct that holds the TypedStructor options,
fields and parameters.

TypedStructor.Plugin behaviour
    



      
This module defines the plugin behaviour for TypedStructor.

  
    
  
  Plugin Behaviour


A plugin is a module that implements the TypedStructor.Plugin behaviour.
Three macro callbacks are available for injecting code at different stages:
	init/1: This macro callback is called when the plugin is used.
	before_definition/2: This macro callback is called right before defining the struct.
Note hat plugins will run in the order they are registered.
	after_definition/2: This macro callback is called right after defining the struct.
Note that plugins will run in the reverse order they are registered.

Plugins guides
Here are some Plugin Guides
for creating your own plugins. Please check them out
and feel free to copy-paste the code.


  
    
  
  Example


Let's define a plugin that defines Ecto.Schema while defining a typed struct.
This plugin takes a :source option which passing to Ecto.Schema.schema/2,
you can use belongs_to and has_many directly in the module.
It would be used like this:
defmodule MyApp do
  use TypedStructor

  # fix aliases
  alias __MODULE__.User
  alias __MODULE__.Post

  typed_structor module: User do
    # import the plugin with source option
    plugin EctoSchemaPlugin, source: "users"

    field :name, :string, enforce: true
    field :age, :integer, default: 0
    # pass redact option to Ecto.Schema.field/3
    field :password, :string, redact: true
    has_many :posts, Post
  end

  typed_structor module: Post do
    # import the plugin with source option
    plugin EctoSchemaPlugin, source: "posts"

    field :title, :string, enforce: true
    field :content, :string, enforce: true
    belongs_to :user, User, enforce: true
  end
end
After compiled, you got:
iex> t MyApp.User
@type t() :: %MyApp.User{
        __meta__: Ecto.Schema.Metadata.t(),
        age: integer() | nil,
        id: integer(),
        name: String.t(),
        password: String.t() | nil,
        posts: [MyApp.Post.t()] | nil
      }

iex> t MyApp.Post
@type t() :: %MyApp.Post{
        __meta__: Ecto.Schema.Metadata.t(),
        content: String.t(),
        id: integer(),
        title: String.t(),
        user: MyApp.User.t(),
        user_id: integer()
      }

iex> MyApp.User.__schema__(:redact_fields)
[:password]

iex> MyApp.User.__schema__(:association, :posts)
%Ecto.Association.Has{
  cardinality: :many,
  field: :posts,
  owner: MyApp.User,
  related: MyApp.Post,
  owner_key: :id,
  related_key: :user_id,
  on_cast: nil,
  queryable: MyApp.Post,
  on_delete: :nothing,
  on_replace: :raise,
  where: [],
  unique: true,
  defaults: [],
  relationship: :child,
  ordered: false,
  preload_order: []
}

iex> MyApp.Post.__schema__(:association, :user)
%Ecto.Association.BelongsTo{
  field: :user,
  owner: MyApp.Post,
  related: MyApp.User,
  owner_key: :user_id,
  related_key: :id,
  queryable: MyApp.User,
  on_cast: nil,
  on_replace: :raise,
  where: [],
  defaults: [],
  cardinality: :one,
  relationship: :parent,
  unique: true,
  ordered: false
}
Following is the implementation of the plugin:
defmodule EctoSchemaPlugin do
  use TypedStructor.Plugin

  @impl TypedStructor.Plugin
  defmacro init(opts) do
    quote do
      unless Keyword.has_key?(unquote(opts), :source) do
        raise "The `:source` option is not provided."
      end

      # import association functions to the module,
      # so that we can use `has_many` and `belongs_to` directly
      import unquote(__MODULE__), only: [has_many: 2, belongs_to: 3]
    end
  end

  @impl TypedStructor.Plugin
  defmacro before_definition(definition, _opts) do
    # manipulate the definition before defining the struct
    quote do
      unquote(definition)
      # disable defining struct, for Ecto.Schema will define it
      |> Map.update!(:options, &Keyword.put(&1, :define_struct, false))
      |> Map.update!(:fields, fn fields ->
        Enum.flat_map(fields, fn field ->
          {ecto_type, options} = Keyword.pop!(field, :type)
          type = unquote(__MODULE__).__ecto_type_to_type__(ecto_type)

          field = Keyword.merge(options, type: type, ecto_type: ecto_type)

          case ecto_type do
            {:belongs_to, name} ->
              foreign_key_name =
                name
                |> Macro.expand(__ENV__)
                |> Module.split()
                |> List.last()
                |> Macro.underscore()
                |> Kernel.<>("_id")
                |> String.to_atom()

              foreign_key =
                Keyword.merge(options, name: foreign_key_name, type: quote(do: integer()))

              [foreign_key, field]

            _other ->
              [field]
          end
        end)
      end)
      |> Map.update!(
        :fields,
        &[
          [name: :__meta__, type: quote(do: Ecto.Schema.Metadata.t()), enforce: true],
          [name: :id, type: quote(do: integer()), enforce: true]
          | &1
        ]
      )
    end
  end

  @impl TypedStructor.Plugin
  defmacro after_definition(definition, opts) do
    # here we define the Ecto.Schema
    quote bind_quoted: [definition: definition, opts: opts] do
      use Ecto.Schema

      source = Keyword.fetch!(opts, :source)

      schema source do
        for options <- definition.fields do
          {name, options} = Keyword.pop!(options, :name)
          {ecto_type, options} = Keyword.pop(options, :ecto_type)
          options = Keyword.take(options, [:primary_key, :default, :redact])

          case ecto_type do
            nil ->
              # skip some fields
              nil

            {:has_many, module} ->
              module = Macro.expand(module, __ENV__)

              has_many name, module, options

            {:belongs_to, module} ->
              module = Macro.expand(module, __ENV__)

              belongs_to name, module, options

            _ ->
              field name, ecto_type, options
          end
        end
      end
    end
  end

  defmacro has_many(name, queryable) do
    quote do
      field unquote(name), {:has_many, unquote(queryable)}
    end
  end

  defmacro belongs_to(name, queryable, opts) do
    quote do
      field unquote(name), {:belongs_to, unquote(queryable)}, unquote(opts)
    end
  end

  def __ecto_type_to_type__(:string), do: quote(do: String.t())
  def __ecto_type_to_type__(:integer), do: quote(do: integer())
  def __ecto_type_to_type__({:has_many, module}), do: quote(do: [unquote(module).t()])
  def __ecto_type_to_type__({:belongs_to, module}), do: quote(do: unquote(module).t())
end

